

# Stabilized Optimization Via an NCL Algorithm



Ding Ma, Kenneth L. Judd, Dominique Orban  
and Michael A. Saunders

**Abstract** For optimization problems involving many nonlinear inequality constraints, we extend the bound-constrained (BCL) and linearly constrained (LCL) augmented Lagrangian approaches of LANCELOT and MINOS to an algorithm that solves a sequence of nonlinearly constrained augmented Lagrangian subproblems whose nonlinear constraints satisfy the LICQ everywhere. The NCL algorithm is implemented in AMPL and tested on large instances of a tax policy model that could not be solved directly by the state-of-the-art solvers that we tested, because of singularity in the Jacobian of the active constraints. Algorithm NCL with IPOPT as subproblem solver proves to be effective, with IPOPT using second derivatives and successfully warm starting each subproblem.

**Keywords** Stabilized optimization · LICQ · Augmented Lagrangian · BCL  
NCL · Interior method · Warm start

---

D. Ma  
Management Science and Engineering,  
Stanford University, Stanford, CA, USA  
e-mail: dingma@stanford.edu

K. L. Judd  
Hoover Institution, Stanford University, Stanford, CA, USA  
e-mail: judd@stanford.edu

D. Orban  
GERAD and Department of Mathematics and Industrial Engineering,  
École Polytechnique, Montréal, QC, Canada  
e-mail: dominique.orban@gerad.ca

M. A. Saunders (✉)  
Department of Management Science and Engineering,  
Stanford University, Stanford, CA, USA  
e-mail: saunders@stanford.edu

## 1 Introduction

We consider constrained optimization problems of the form

NCO	$\begin{aligned} & \underset{x \in \mathbb{R}^n}{\text{minimize}} && \phi(x) \\ & \text{subject to} && c(x) \geq 0, \quad Ax \geq b, \quad \ell \leq x \leq u, \end{aligned}$
-----	---

where  $\phi(x)$  is a smooth nonlinear function,  $c(x) \in \mathbb{R}^m$  is a vector of smooth nonlinear functions, and  $Ax \geq b$  is a placeholder for a set of linear inequality or equality constraints, with  $x$  lying between lower and upper bounds  $\ell$  and  $u$ .

In some applications where  $m \gg n$ , there may be more than  $n$  constraints that are essentially active at a solution. The constraints do not satisfy the linear independence constraint qualification (LICQ), and general-purpose solvers are likely to have difficulty converging. Some form of regularization is required. The stabilized SQP methods of Wright [20] and Gill et al. [9, 10] have been developed specifically for such problems. We achieve reliability more simply by adapting the augmented Lagrangian algorithm of the general-purpose optimization solver LANCELOT [4, 5, 15] in the vein of Arreckx and Orban [2] to derive a sequence of regularized subproblems denoted in the next section by  $\text{NC}_k$ .

## 2 BCL, LCL, and NCL Methods

The theory for the large-scale solver LANCELOT is best described in terms of the general optimization problem

NECB	$\begin{aligned} & \underset{x \in \mathbb{R}^n}{\text{minimize}} && \phi(x) \\ & \text{subject to} && c(x) = 0, \quad \ell \leq x \leq u \end{aligned}$
------	--

with *nonlinear equality constraints* and bounds. We let  $x^*$  denote a local solution of NECB and  $(y^*, z^*)$  denote associated multipliers. LANCELOT treats NECB by solving a sequence of *bound-constrained subproblems* of the form

$\text{BC}_k$	$\begin{aligned} & \underset{x}{\text{minimize}} && L(x, y_k, \rho_k) = \phi(x) - y_k^T c(x) + \frac{1}{2} \rho_k \ c(x)\ ^2 \\ & \text{subject to} && \ell \leq x \leq u, \end{aligned}$
---------------	---

where  $y_k$  is an estimate of the Lagrange multipliers  $y^*$  for the equality constraints. This was called a bound-constrained Lagrangian (BCL) method by Friedlander and Saunders [8], in contrast to the linearly constrained Lagrangian methods (LCL) of Robinson [18] and MINOS [16], whose subproblems  $\text{LC}_k$  contain bounds as in  $\text{BC}_k$

and also linearizations of the equality constraints at the current point  $x_k$  (including linear constraints).

In order to treat NCO with a sequence of  $BC_k$  subproblems, we convert the nonlinear inequality constraints to equalities to obtain

$$\begin{array}{ll} \text{NCO}' & \underset{x, s}{\text{minimize}} \quad \phi(x) \\ & \text{subject to} \quad c(x) - s = 0, \quad Ax \geq b, \quad \ell \leq x \leq u, \quad s \geq 0 \end{array}$$

with corresponding subproblems (including linear constraints)

$$\begin{array}{ll} \text{BC}'_k & \underset{x, s}{\text{minimize}} \quad L(x, y_k, \rho_k) = \phi(x) - y_k^T(c(x) - s) + \frac{1}{2}\rho_k \|c(x) - s\|^2 \\ & \text{subject to} \quad Ax \geq b, \quad \ell \leq x \leq u, \quad s \geq 0. \end{array}$$

We now introduce variables  $r = -(c(x) - s)$  into  $BC'_k$  to obtain the *nonlinearly constrained Lagrangian* (NCL) subproblem

$$\begin{array}{ll} \text{NC}_k & \underset{x, r}{\text{minimize}} \quad \phi(x) + y_k^T r + \frac{1}{2}\rho_k \|r\|^2 \\ & \text{subject to} \quad c(x) + r \geq 0, \quad Ax \geq b, \quad \ell \leq x \leq u, \end{array}$$

in which  $r$  serves to make the nonlinear constraints independent. (If NCO includes an equality  $c_i(x) = 0$ ,  $\text{NC}_k$  would contain  $c_i(x) + r_i = 0$ .) Assuming existence of finite multipliers and feasibility, for  $\rho_k > 0$  and larger than a certain finite value, the NCL subproblems should cause  $y_k$  to approach  $y^*$  and most of the solution  $(x_k^*, r_k^*, y_k^*, z_k^*)$  of  $\text{NC}_k$  to approach  $(x^*, y^*, z^*)$ , with  $r_k^*$  approaching zero.

Problem  $\text{NC}_k$  is analogous to Friedlander and Orban’s formulation for convex quadratic programs [7, Eq. (3.2)]. See also Arreckx and Orban [2], where the motivation is the same as here, achieving reliability when the nonlinear constraints do not satisfy LICQ.

Note that for general problems NECB, the BCL and LCL subproblems contain linear constraints (bounds only, or linearized constraints and bounds). Our NCL formulation retains nonlinear constraints in the  $\text{NC}_k$  subproblems, but simplifies them by ensuring that they satisfy LICQ. On large problems, the additional variables  $r \in \mathbb{R}^m$  in  $\text{NC}_k$  may be detrimental to active-set solvers like MINOS or SNOPT [11] because they increase the number of degrees of freedom (superbasic variables). Fortunately, they are easily accommodated by interior methods, as our numerical results show for IPOPT [12, 19]. We expect the same to be true for KNITRO [3, 14]. These solvers are most effective when second derivatives are available, as they are for our AMPL model.

## 2.1 The BCL Algorithm

The LANCELOT BCL method is summarized in Algorithm BCL. Each subproblem  $BC_k$  is solved with a specified optimality tolerance  $\omega_k$ , generating an iterate  $x_k^*$  and the associated Lagrangian gradient  $z_k^* \equiv \nabla L(x_k^*, y_k, \rho_k)$ . If  $\|c(x_k^*)\|$  is sufficiently small, the iteration is regarded as “successful” and an update to  $y_k$  is computed from  $x_k^*$ . Otherwise,  $y_k$  is not altered but  $\rho_k$  is increased.

Key properties are that the subproblems are solved inexactly, the penalty parameter is increased only finitely often, and the multiplier estimates  $y_k$  need not be assumed bounded. Under certain conditions, all iterations are eventually successful, the  $\rho_k$ 's remain constant, the iterates converge superlinearly, and the algorithm terminates in a finite number of iterations [4].

---

### Algorithm 1 BCL (Bound-Constrained Lagrangian Method for NECB)

---

```

1: procedure BCL( $x_0, y_0, z_0$ )
2:   Set penalty parameter  $\rho_1 > 0$ , scale factor  $\tau > 1$ , and constants  $\alpha, \beta > 0$  with  $\alpha < 1$ .
3:   Set positive convergence tolerances  $\eta_*, \omega_* \ll 1$  and infeasibility tolerance  $\eta_1 > \eta_*$ .
4:    $k \leftarrow 0$ , converged  $\leftarrow$  false
5:   repeat
6:      $k \leftarrow k + 1$ 
7:     Choose optimality tolerance  $\omega_k > 0$  such that  $\lim_{k \rightarrow \infty} \omega_k \leq \omega_*$ .
8:     Find  $(x_k^*, z_k^*)$  that solves  $BC_k$  to within  $\omega_k$ .
9:     if  $\|c(x_k^*)\| \leq \max(\eta_*, \eta_k)$  then
10:       $y_k^* \leftarrow y_k - \rho_k c(x_k^*)$ 
11:       $x_k \leftarrow x_k^*, y_k \leftarrow y_k^*, z_k \leftarrow z_k^*$  update solution estimates
12:      if  $(x_k, y_k, z_k)$  solves NECB to within  $\omega_*$ , converged  $\leftarrow$  true
13:       $\rho_{k+1} \leftarrow \rho_k$  keep  $\rho_k$ 
14:       $\eta_{k+1} \leftarrow \eta_k / (1 + \rho_{k+1}^\beta)$  decrease  $\eta_k$ 
15:    else
16:       $\rho_{k+1} \leftarrow \tau \rho_k$  increase  $\rho_k$ 
17:       $\eta_{k+1} \leftarrow \eta_0 / (1 + \rho_{k+1}^\alpha)$  may increase or decrease  $\eta_k$ 
18:    end if
19:  until converged
20:   $x^* \leftarrow x_k, y^* \leftarrow y_k, z^* \leftarrow z_k$ 
21: end procedure

```

---

Note that at step 8 of Algorithm BCL, the inexact minimization would typically use the initial guess  $(x_k^*, z_k^*)$ . However, other initial points are possible. At step 12, we say that  $(x_k, y_k, z_k)$  solves NECB to within  $\omega_*$  if the largest dual infeasibility is smaller than  $\omega_*$ .

**Algorithm 2** NCL (Nonlinearly Constrained Lagrangian Method for NCO)

---

```

1: procedure NCL( $x_0, r_0, y_0, z_0$ )
2:   Set penalty parameter  $\rho_1 > 0$ , scale factor  $\tau > 1$ , and constants  $\alpha, \beta > 0$  with  $\alpha < 1$ .
3:   Set positive convergence tolerances  $\eta_*, \omega_* \ll 1$  and infeasibility tolerance  $\eta_1 > \eta_*$ .
4:    $k \leftarrow 0$ , converged  $\leftarrow$  false
5:   repeat
6:      $k \leftarrow k + 1$ 
7:     Choose optimality tolerance  $\omega_k > 0$  such that  $\lim_{k \rightarrow \infty} \omega_k \leq \omega_*$ .
8:     Find  $(x_k^*, r_k^*, y_k^*, z_k^*)$  that solves  $\text{NC}_k$  to within  $\omega_k$ .
9:     if  $\|r_k^*\| \leq \max(\eta_*, \eta_k)$  then
10:       $y_k^* \leftarrow y_k + \rho_k r_k^*$ 
11:       $x_k \leftarrow x_k^*, r_k \leftarrow r_k^*, y_k \leftarrow y_k^*, z_k \leftarrow z_k^*$  update solution estimates
12:      if  $(x_k, y_k, z_k)$  solves NCO to within  $\omega_*$ , converged  $\leftarrow$  true
13:       $\rho_{k+1} \leftarrow \rho_k$  keep  $\rho_k$ 
14:       $\eta_{k+1} \leftarrow \eta_k / (1 + \rho_{k+1}^\beta)$  decrease  $\eta_k$ 
15:    else
16:       $\rho_{k+1} \leftarrow \tau \rho_k$  increase  $\rho_k$ 
17:       $\eta_{k+1} \leftarrow \eta_0 / (1 + \rho_{k+1}^\alpha)$  may increase or decrease  $\eta_k$ 
18:    end if
19:  until converged
20:   $x^* \leftarrow x_k, r^* \leftarrow r_k, y^* \leftarrow y_k, z^* \leftarrow z_k$ 
21: end procedure

```

---

## 2.2 The NCL Algorithm

To derive a stabilized algorithm for problem NCO, we modify Algorithm BCL by introducing  $r$  and replacing the subproblems  $\text{BC}_k$  by  $\text{NC}_k$ . The resulting method is summarized in Algorithm NCL. The update to  $y_k$  becomes  $y_k^* \leftarrow y_k - \rho_k(c(x_k^*) - s_k^*) = y_k + \rho_k r_k^*$ , the value satisfied by an optimal  $y_k^*$  for subproblem  $\text{NC}_k$ . Step 8 of Algorithm NCL would typically use  $(x_k^*, r_k^*, y_k^*, z_k^*)$  as initial guess, and that is what we use in our implementation below.

## 3 An Application: Optimal Tax Policy

Some challenging test cases arise from the tax policy models described in [13]. With  $x = (c, y)$ , they take the form

TAX	maximize <sub><math>c, y</math></sub> $\sum_i \lambda_i U^i(c_i, y_i)$ subject to $U^i(c_i, y_i) - U^i(c_j, y_j) \geq 0$ for all $i, j$ $\lambda^T(y - c) \geq 0$ $c, y \geq 0,$
-----	---

where  $c_i$  and  $y_i$  are the consumption and income of taxpayer  $i$ , and  $\lambda$  is a vector of positive weights. The utility functions  $U^i(c_i, y_i)$  are each of the form

$$U(c, y) = \frac{(c - \alpha)^{1-1/\gamma}}{1 - 1/\gamma} - \psi \frac{(y/w)^{1/\eta+1}}{1/\eta + 1},$$

where  $w$  is the wage rate and  $\alpha, \gamma, \psi,$  and  $\eta$  are taxpayer heterogeneities. More precisely, the utility functions are of the form

$$U^{i,j,k,g,h}(c_{p,q,r,s,t}, y_{p,q,r,s,t}) = \frac{(c_{p,q,r,s,t} - \alpha_k)^{1-1/\gamma_h}}{1 - 1/\gamma_h} - \psi_g \frac{(y_{p,q,r,s,t}/w_i)^{1/\eta_j+1}}{1/\eta_j + 1},$$

where  $(i, j, k, g, h)$  and  $(p, q, r, s, t)$  run over  $na$  wage types,  $nb$  elasticities of labor supply,  $nc$  basic need types,  $nd$  levels of distaste for work, and  $ne$  elasticities of demand for consumption, with  $na, nb, nc, nd, ne$  determining the size of the problem, namely  $m = T(T - 1)$  nonlinear constraints,  $n = 2T$  variables, with  $T := na \times nb \times nc \times nd \times ne$ .

Table 1 summarizes results for a 4D example ( $ne = 1$  and  $\gamma_1 = 1$ ). The first term of  $U(c, y)$  becomes  $\log(c - \alpha)$ , the limit as  $\gamma \rightarrow 1$ . Problem NCO and Algorithm NCL were formulated in the AMPL modeling language [6]. The solvers SNOPT [11] and IPOPT [19] were unable to solve NCO itself, but Algorithm NCL was successful with IPOPT solving the subproblems  $NC_k$ . We use a default configuration of IPOPT with MUMPS [1] as symmetric indefinite solver to compute search directions. We set the optimality tolerance for IPOPT to  $\omega_k = \omega_* = 10^{-6}$  throughout and specified warm starts for  $k \geq 2$  using options `warm_start_init_point=yes` and `mu_init=1e-4`. These options greatly improved the performance of IPOPT on each subproblem compared to cold starts, for which `mu_init=0.1`. It is helpful that only the objective function of  $NC_k$  changes with  $k$ .

**Table 1** NCL results on a 4D example with  $na, nb, nc, nd = 11, 3, 3, 2$ , giving  $m = 39006, n = 395$ . Itns refers to IPOPT’s primal-dual interior point method, and time is seconds on an Apple iMac with 2.93 GHz Intel Core i7

$k$	$\rho_k$	$\eta_k$	$\ r_k^*\ _\infty$	$\phi(x_k^*)$	Itns	Time
1	$10^2$	$10^{-2}$	$3.1e - 03$	$-2.1478532e + 01$	125	42.8
2	$10^2$	$10^{-3}$	$1.3e - 03$	$-2.1277587e + 01$	18	6.5
3	$10^3$	$10^{-3}$	$6.6e - 04$	$-2.1177152e + 01$	27	9.1
4	$10^3$	$10^{-4}$	$5.5e - 04$	$-2.1110210e + 01$	31	10.8
5	$10^4$	$10^{-4}$	$2.9e - 04$	$-2.1066664e + 01$	57	24.3
6	$10^5$	$10^{-4}$	$6.5e - 05$	$-2.1027152e + 01$	75	26.8
7	$10^5$	$10^{-5}$	$5.2e - 05$	$-2.1018896e + 01$	130	60.9
8	$10^6$	$10^{-5}$	$9.3e - 06$	$-2.1015295e + 01$	159	81.8
9	$10^6$	$10^{-6}$	$2.0e - 06$	$-2.1014808e + 01$	139	70.0
10	$10^7$	$10^{-6}$	$2.1e - 07$	$-2.1014800e + 01$	177	97.6

For this example, problem NCO has  $m = 39,006$  nonlinear inequality constraints and one linear constraint in  $n = 395$  variables  $x = (c, y)$ , and nonnegativity bounds. Subproblem  $NC_k$  has 39,007 constraints and 39,402 variables when  $r$  is included. Fortunately,  $r$  does not affect the complexity of each IPOPT iteration, but greatly improves stability. In contrast, active-set methods like MINOS and SNOPT are very inefficient on the  $NC_k$  subproblems because the large number of inequality constraints leads to thousands of minor iterations, and the presence of  $r$  (with no bounds) leads to thousands of superbasic variables. About  $3.2n$  constraints were within  $10^{-6}$  of being active.

Table 2 summarizes results for a 5D example. The  $NC_k$  subproblems have  $m = 32,220$  nonlinear constraints and  $n = 360$  variables, leading to 32,581 variables including  $r$ . Again the options `warm_start_init_point=yes` and `mu_init=1e-4` for  $k \geq 2$  led to good performance by IPOPT on each subproblem. About  $3n$  constraints were within  $10^{-6}$  of being active.

**Table 2** NCL results on a 5D example with  $na, nb, nc, nd, ne = 5, 3, 3, 2, 2$ , giving  $m = 32220$ ,  $n = 360$

$k$	$\rho_k$	$\eta_k$	$\ r_k^*\ _\infty$	$\phi(x_k^*)$	Itns	Time
1	$10^2$	$10^{-2}$	$7.0e - 03$	$-4.2038075e + 02$	95	41.1
2	$10^2$	$10^{-3}$	$4.1e - 03$	$-4.2002898e + 02$	17	7.2
3	$10^3$	$10^{-3}$	$1.3e - 03$	$-4.1986069e + 02$	20	8.1
4	$10^4$	$10^{-3}$	$4.4e - 04$	$-4.1972958e + 02$	48	25.0
5	$10^4$	$10^{-4}$	$2.2e - 04$	$-4.1968646e + 02$	43	20.5
6	$10^5$	$10^{-4}$	$9.8e - 05$	$-4.1967560e + 02$	64	32.9
7	$10^5$	$10^{-5}$	$6.6e - 05$	$-4.1967177e + 02$	57	26.8
8	$10^6$	$10^{-5}$	$4.2e - 06$	$-4.1967150e + 02$	87	46.2
9	$10^6$	$10^{-6}$	$9.4e - 07$	$-4.1967138e + 02$	96	53.6

**Table 3** NCL results on a 5D example with  $na, nb, nc, ne, ne = 21, 3, 3, 2, 2$ , giving  $m = 570780$ ,  $n = 1512$

$k$	$\rho_k$	$\eta_k$	$\ r_k^*\ _\infty$	$\phi(x_k^*)$	mu_init	Itns	Time
1	$10^2$	$10^{-2}$	$5.1e - 03$	$-1.7656816e + 03$	$10^{-1}$	825	7763.3
2	$10^2$	$10^{-3}$	$2.4e - 03$	$-1.7648480e + 03$	$10^{-4}$	66	472.8
3	$10^3$	$10^{-3}$	$1.3e - 03$	$-1.7644006e + 03$	$10^{-4}$	106	771.3
4	$10^4$	$10^{-3}$	$3.8e - 04$	$-1.7639491e + 03$	$10^{-5}$	132	1347.0
5	$10^4$	$10^{-4}$	$3.2e - 04$	$-1.7637742e + 03$	$10^{-5}$	229	2450.9
6	$10^5$	$10^{-4}$	$8.6e - 05$	$-1.7636804e + 03$	$10^{-6}$	104	1096.9
7	$10^5$	$10^{-5}$	$4.9e - 05$	$-1.7636469e + 03$	$10^{-6}$	143	1633.4
8	$10^6$	$10^{-5}$	$1.5e - 05$	$-1.7636252e + 03$	$10^{-7}$	71	786.1
9	$10^7$	$10^{-5}$	$2.8e - 06$	$-1.7636196e + 03$	$10^{-7}$	67	725.7
10	$10^7$	$10^{-6}$	$5.1e - 07$	$-1.7636187e + 03$	$10^{-8}$	18	171.0

For much larger problems of this type, we found that it was helpful to reduce `mu_init` more often, as illustrated in Table 3. The  $\text{NC}_k$  subproblems here have  $m = 570, 780$  nonlinear constraints and  $n = 1512$  variables, leading to 572,292 variables including  $r$ . Note that the number of NCL iterations is stable ( $k \leq 10$ ), and IPOPT performs well on each subproblem with decreasing `mu_init`. This time about  $6.6n$  constraints were within  $10^{-6}$  of being active.

Note that the LANCELOT approach allows early subproblems to be solved less accurately [4]. It may save time to set  $\omega_k = \eta_k$  (say) rather than  $\omega_k = \omega_*$  throughout.

## 4 Conclusions

This work has been illuminating in several ways as we sought to improve our ability to solve examples of problem TAX.

- Small examples of the tax model solve efficiently with MINOS and SNOPT, but eventually fail to converge as the problem size increases.
- IPOPT also solves small examples efficiently, but eventually starts requesting additional memory for the MUMPS sparse linear solver. The solver may freeze, or the iterations may diverge.
- The  $\text{NC}_k$  subproblems are not suitable for MINOS or SNOPT because of the large number of variables ( $x, r$ ) and the resulting number of superbasic variables (although warm starts are natural).
- It is often said that interior methods cannot be warm started. Nevertheless, IPOPT has several runtime options that have proved to be extremely helpful for implementing Algorithm NCL. For the results obtained here, it has been sufficient to say that warm starts are wanted for  $k > 1$ , and that the IPOPT barrier parameter should be initialized at decreasing values for later  $k$  (where only the objective of subproblem  $\text{NC}_k$  changes with  $k$ ).
- The numerical examples of Sect. 3 had  $3n, 3n$ , and  $6.6n$  constraints essentially active at the solution, yet were solved successfully. They suggest that the NCL approach with an interior method as subproblem solver can overcome LICQ difficulties on problems that could not be solved directly.

### Funding

This work was supported by the National Institute of General Medical Sciences of the National Institutes of Health [award U01GM102098] (DM and MAS) and an NSERC Discovery Grant (DO).

**Acknowledgements** We are extremely grateful to the developers of AMPL and IPOPT for making the development and evaluation of Algorithm NCL possible. We are especially grateful to Mehiddin Al-Baali and other organizers of the NAO-IV conference *Numerical Analysis and Optimization* at Sultan Qaboos University, Muscat, Oman, which brought the authors and AMPL developers together in January 2017. We also thank the reviewer for final helpful suggestions.



## Appendix A AMPL Models, Data, and Scripts

Algorithm NCL has been implemented in the AMPL modeling language [6] and tested on problem TAX. The following sections list each relevant file. The files are available from [17].

### A.1 Tax Model

File `pTax5Dncl.mod` codes subproblem  $NC_k$  for problem TAX with five parameters  $w, \eta, \alpha, \psi, \gamma$ , using  $\mu := 1/\eta$ . Note that for  $U(c, y)$  in the objective and constraint functions, the first term  $(c - \alpha)^{1-1/\gamma}/(1 - 1/\gamma)$  is replaced by a piecewise-smooth function that is defined for all values of  $c$  and  $\alpha$  (see [13]).

Primal regularization  $\frac{1}{2}\delta\|(c, y)\|^2$  with  $\delta = 10^{-8}$  is added to the objective function to promote uniqueness of the minimizer. The vector  $r$  is called R to avoid a clash with subscript  $r$ .

---

```

1 # pTax5Dncl.mod
2
3 # Define parameters for agents (taxpayers)
4 param na;           # number of types in wage
5 param nb;           # number of types in eta
6 param nc;           # number of types in alpha
7 param nd;           # number of types in psi
8 param ne;           # number of types in gamma
9 set A := 1..na;     # set of wages
10 set B := 1..nb;    # set of eta
11 set C := 1..nc;    # set of alpha
12 set D := 1..nd;    # set of psi
13 set E := 1..ne;    # set of gamma
14 set T = {A,B,C,D,E}; # set of agents
15
16 # Define wages for agents (taxpayers)
17 param wmin;        # minimum wage level
18 param wmax;        # maximum wage level
19 param w {A};       # i, wage vector
20 param mu {B};      # j, mu = 1/eta# mu vector
21 param mu1 {B};     # mu1[j] = mu[j] + 1
22 param alpha {C};   # k, ak vector for utility
23 param psi {D};     # g
24 param gamma {E};   # h
25 param lambda {A,B,C,D,E}; # distribution density
26 param epsilon;
27 param primreg      default 1e-8; # Small primal
28                                # regularization
29
30 var y{(i,j,k,g,h) in T} >= 0.1; # consumption for
31                                # tax payer
32                                # (i,j,k,g,h)

```

```

33 var y{(i,j,k,g,h) in T} >= 0.1; # income for
34     tax payer
35     (i,j,k,g,h)
36 var R{(i,j,k,g,h) in T, (p,q,r,s,t) in T:
37     !(i=p and j=q and k=r and g=s and h=t)}
38     >= -1e+20, <= 1e+20;
39
40 param kmax      default 20; # limit on
41     NCL itns
42 param rhok      default 1e+2; # augmented
43     Lagrangian penalty
44     parameter
45 param rhofac    default 10.0; # increase
46     factor
47 param rhomax    default 1e+8; # biggest rhok
48 param etak      default 1e-2; # opttol for
49     augmented
50     Lagrangian loop
51 param etafac    default 0.1; # reduction factor for
52     opttol
53 param etamin    default 1e-8; # smallest etak
54 param rmax      default 0; # max r (for printing)
55
56 param rmin      default 0; # min r (for printing)
57
58 param rnorm     default 0; # ||r||_inf
59 param rtol      default 1e-6; # quit if biggest
60     |r_i| <= rtol
61
62 param nT        default 1; # nT = na*nb
63     *nc*nd*ne
64 param m         default 1; # nT*(nT-1)
65     = no. of nonlinear
66     constraints
67 param n         default 1; # 2*nT
68     = no. of nonlinear
69     variables
70
71 param ck{(i,j,k,g,h) in T} default 0;
72     # current variable c
73 param yk{(i,j,k,g,h) in T} default 0;
74     # current variable y
75 param rk{(i,j,k,g,h) in T, (p,q,r,s,t) in T:
76     # current variable r = - (c(x) - s)
77     !(i=p and j=q and k=r and g=s and h=t)} default 0;
78 param dk{(i,j,k,g,h) in T, (p,q,r,s,t) in T:
79     # current dual variables (y_k)
80     !(i=p and j=q and k=r and g=s and h=t)} default 0;
81
82 minimize f:
83     sum{(i,j,k,g,h) in T}

```

```

84      (
85          (if c[i,j,k,g,h] - alpha[k] >= epsilon then
86              - lambda[i,j,k,g,h] *
87                  ((c[i,j,k,g,h] - alpha[k])
88                     ^ (1-1/gamma[h]) / (1-1/gamma[h])
89                     - psi[g]*(y[i,j,k,g,h]/w[i])
90                     ^mul[j] / mul[j])
91          else
92              - lambda[i,j,k,g,h] *
93                  (- 0.5/gamma[h] * epsilon^(-1/gamma[h]-1)
94                   * (c[i,j,k,g,h] - alpha[k])^2
95                   + (1+1/gamma[h])*epsilon^(-1/gamma[h])
96                   * (c[i,j,k,g,h] - alpha[k])
97                   + (1/(1-1/gamma[h]) - 1 - 0.5/gamma[h])
98                   * epsilon^(1-1/gamma[h])
99                   - psi[g]*(y[i,j,k,g,h]/w[i])
100                  ^mul[j] / mul[j])
101          )
102      + 0.5 * primreg * (c[i,j,k,g,h]^2
103      + y[i,j,k,g,h]^2)
104      )
105  + sum{(i,j,k,g,h) in T, (p,q,r,s,t) in
106  T: !(i=p and j=q and k=r and g=s and h=t)}
107      (dk[i,j,k,g,h,p,q,r,s,t]
108       * R[i,j,k,g,h,p,q,r,s,t]
109       + 0.5 * rhok
110       * R[i,j,k,g,h,p,q,r,s,t]^2);
111
112  subject to
113
114  Incentive{(i,j,k,g,h) in T, (p,q,r,s,t) in T:
115      !(i=p and j=q and k=r and g=s and h=t)}:
116      (if c[i,j,k,g,h] - alpha[k] >= epsilon then
117          (c[i,j,k,g,h] - alpha[k])
118          ^ (1-1/gamma[h]) / (1-1/gamma[h])
119          - psi[g]*(y[i,j,k,g,h]/w[i])
120          ^mul[j] / mul[j]
121      else
122          - 0.5/gamma[h] * epsilon^(-1/gamma[h]-1)
123          * (c[i,j,k,g,h] - alpha[k])^2
124          + (1+1/gamma[h])*epsilon^(-1/gamma[h])
125          * (c[i,j,k,g,h] - alpha[k])
126          + (1/(1-1/gamma[h]) - 1 - 0.5/gamma[h])
127          * epsilon^(1-1/gamma[h])
128          - psi[g]*(y[i,j,k,g,h]/w[i])^mul[j]
129          / mul[j]
130      )
131      - (if c[p,q,r,s,t] - alpha[k] >= epsilon then
132          (c[p,q,r,s,t] - alpha[k])^(1-1/gamma[h])
133          / (1-1/gamma[h])
134          - psi[g]*(y[p,q,r,s,t]/w[i])^mul[j] / mul[j]

```

```

135     else
136         - 0.5/gamma[h] *epsilon^(-1/gamma[h]-1)
137         *(c[p,q,r,s,t] - alpha[k])^2
138         + (1+1/gamma[h])*epsilon^(-1/gamma[h] )
139         *(c[p,q,r,s,t] - alpha[k])
140         + (1/(1-1/gamma[h]) - 1 - 0.5/gamma[h])
141         *epsilon^(1-1/gamma[h])
142         - psi[g]*(y[p,q,r,s,t]/w[i])^mul[j] / mul[j]
143     )
144 + R[i,j,k,g,h,p,q,r,s,t] >= 0;
145
146 Technology:
147     sum{(i,j,k,g,h) in T} lambda[i,j,k,g,h]
148     *(y[i,j,k,g,h] - c[i,j,k,g,h]) >= 0;

```

---

## A.2 Tax Model Data

File pTax5Dnc1.dat provides data for a specific problem.

---

```

1 # pTax5Dnc1.dat
2
3 data;
4
5 let na := 5;
6 let nb := 3;
7 let nc := 3;
8 let nd := 2;
9 let ne := 2;
10
11 # Set up wage dimension intervals
12 let wmin := 2;
13 let wmax := 4;
14 let {i in A} w[i] := wmin + ((wmax-wmin)
15 / (na-1)) * (i-1);
16
17 data;
18
19 param mu :=
20     1  0.5
21     2  1
22     3  2 ;
23
24 # Define mul
25 let {j in B} mul[j] := mu[j] + 1;
26
27 data;
28
29 param alpha :=

```

```

30     1    0
31     2    1
32     3    1.5;
33
34 param psi :=
35     1    1
36     2    1.5;
37
38 param gamma :=
39     1    2
40     2    3;
41
42 # Set up 5 dimensional distribution
43 let {(i,j,k,g,h) in T} lambda[i,j,k,g,h] := 1;
44
45 # Choose a reasonable epsilon
46 let epsilon := 0.1;

```

---

### A.3 Initial Values

File `pTax5Dinitial.run` solves a simplified model to compute starting values for Algorithm NCL. The nonlinear inequality constraints are removed, and  $y = c$  is enforced. This model solves easily with MINOS or SNOPT on all cases tried. Solution values are output to file `p5Dinitial.dat`.

---

```

1 # pTax5Dinitial.run
2
3 # Define parameters for agents (taxpayers)
4 param na := 5;           # number of types in wage
5 param nb := 3;           # number of types in eta
6 param nc := 3;           # number of types in alpha
7 param nd := 2;           # number of types in psi
8 param ne := 2;           # number of types in gamma
9 set A := 1..na;          # set of wages
10 set B := 1..nb;          # set of eta
11 set C := 1..nc;          # set of alpha
12 set D := 1..nd;          # set of psi
13 set E := 1..ne;          # set of gamma
14 set T = {A,B,C,D,E};    # set of agents
15
16 # Define wages for agents (taxpayers)
17 param wmin := 2;         # minimum wage level
18 param wmax := 4;         # maximum wage level
19 param w {i in A} := wmin + ((wmax-wmin)
20 / (na-1)) * (i-1);      # wage vector
21
22 # Choose a reasonable epsilon

```

```

23 param epsilon := 0.1;
24
25 # mu vector
26 param mu {B};           # mu = 1/eta
27 param mu1{B};          # mu1[j] = mu[j] + 1
28 param alpha {C};
29 param gamma {E};
30 param psi {D};
31
32 var c {(i,j,k,g,h) in T} >= 0.1;
33 var y {(i,j,k,g,h) in T} >= 0.1;
34
35 maximize f: sum{(i,j,k,g,h) in T}
36   if c[i,j,k,g,h] - alpha[k] >= epsilon then
37     (c[i,j,k,g,h] - alpha[k])^(1-1/gamma[h])
38     / (1-1/gamma[h])
39     - psi[g] * (y[i,j,k,g,h]/w[i])^mu1[j] / mu1[j]
40   else
41     - 0.5/gamma[h] *epsilon^(-1/gamma[h]-1)
42     *(c[i,j,k,g,h] - alpha[k])^2
43     + (1+1/gamma[h])*epsilon^(-1/gamma[h])
44     *(c[i,j,k,g,h] - alpha[k])
45     + (1/(1-1/gamma[h]) - 1 - 0.5/gamma[h])
46     *epsilon^(1-1/gamma[h])
47     - psi[g] * (y[i,j,k,g,h]/w[i])^mu1[j] / mu1[j];
48
49 subject to
50   Budget {(i,j,k,g,h) in T}: y[i,j,k,g,h]
51   - c[i,j,k,g,h] = 0;
52
53 let {(i,j,k,g,h) in T} y[i,j,k,g,h] := i+1;
54 let {(i,j,k,g,h) in T} c[i,j,k,g,h] := i+1;
55
56 data;
57
58 param mu :=
59   1  0.5
60   2  1
61   3  2 ;
62
63 # Define mu1
64 let {j in B} mu1[j] := mu[j] + 1;
65
66 data;
67
68 param alpha :=
69   1  0
70   2  1
71   3  1.5;
72
73 param psi :=

```

```

74      1    1
75      2    1.5;
76
77 param gamma :=
78      1    2
79      2    3;
80
81 option solver snopt;
82 option show_stats 1;
83
84 option snopt_options ' \
85     summary_file=6      \
86     print_file=9       \
87     scale=no           \
88     print_level=0      \
89     major_iterations=2000\
90     iterations=50000   \
91     optimality_tol=1e-7 \
92     *penalty=100.0     \
93     superbasics_limit=3000\
94     solution=yes      \
95     *verify_level=3   \
96 ' ;
97
98
99 display na ,nb ,nc ,nd ,ne;
100 solve;
101 display na ,nb ,nc ,nd ,ne;
102 display y ,c >p5Dinitial.dat;
103 close p5Dinitial.dat;

```

---

## A.4 NCL Implementation

File `pTax5Dnclipopt.run` uses files

```

pTax5Dinitial.run
pTax5Dncl.mod
pTax5Dncl.dat
pTax5Dinitial.dat

```

to implement Algorithm NCL. Subproblems  $NC_k$  are solved in a loop until  $\|r_k^*\|_\infty \leq \text{rtol} = 1e-6$ , or  $\eta_k$  has been reduced to parameter  $\text{etamin} = 1e-8$ , or  $\rho_k$  has been increased to parameter  $\text{rhomax} = 1e+8$ . The loop variable  $k$  is called  $K$  to avoid a clash with subscript  $k$  in the model file. The definitions of  $\text{etak}$  and  $\text{rhok}$  inside the loop are simpler than (but similar to) the settings of  $\eta_k$  and  $\rho_k$  in Algorithm 2.

Optimality tolerance  $\omega_k = \omega_* = 10^{-6}$  is used throughout to ensure that the solution of the final subproblem  $\text{NC}_k$  will be close to a solution of the original problem if  $\|r_k^*\|_\infty$  is small enough for the final  $k$  ( $\|r_k^*\|_\infty \leq \text{rtol} = 1e-6$ ).

IPOPT is used to solve each subproblem  $\text{NC}_k$ , with runtime options set to implement increasingly warm starts.

---

```

1 # pTax5Dnclipopt.run
2
3 reset;    model pTax5Dinitial.run;
4 reset;    model pTax5Dncl.mod;
5 data      pTax5Dncl.dat;
6 data; var include p5Dinitial.dat;
7
8 model;
9 option solver ipopt;
10 option show_stats 1;
11
12 option ipopt_options '\
13   dual_inf_tol=1e-6   \
14   max_iter=5000      \
15   ';
16
17 option opt2 $ipopt_options ' warm_start_init_point
18 =yes';
19
20 # NCL method.
21 # kmax, rhok, rhofac, rhomax, etak, etafac,
22   etamin, rtol
23 # are defined in the .mod file.
24
25 printf "NCLipopt log for pTax5D\n" > 5DNCLipopt.log;
26 display na, nb, nc, nd, ne, primreg > 5DNCLipopt.log;
27 printf "   k      rhok      etak      rnorm
28 Obj\n" > 5DNCLipopt.log;
29
30 for {K in 1..kmax}
31 { display na, nb, nc, nd, ne, primreg, K, kmax,
32   rhok, etak;
33   if K == 2 then {option ipopt_options
34     $opt2 ' mu_init=1e-4'};
35   if K == 4 then {option ipopt_options
36     $opt2 ' mu_init=1e-5'};
37   if K == 6 then {option ipopt_options
38     $opt2 ' mu_init=1e-6'};
39   if K == 8 then {option ipopt_options
40     $opt2 ' mu_init=1e-7'};
41   if K == 10 then {option ipopt_options
42     $opt2 ' mu_init=1e-8'};
43

```



```

44  display $ipopt_options;
45  solve;
46
47  let rmax := max({(i,j,k,g,h) in T, (p,q,r,s,t)
48  in T:
49      !(i=p and j=q and k=r and g=s and h=t)})
50      R[i,j,k,g,h,p,q,r,s,t]);
51  let rmin := min({(i,j,k,g,h) in T, (p,q,r,s,t)
52  in T:
53      !(i=p and j=q and k=r and g=s and h=t)})
54      R[i,j,k,g,h,p,q,r,s,t]);
55  display na, nb, nc, nd, ne, primreg, K, rhok,
56  etak, kmax;
57  display K, kmax, rmax, rmin;
58  let rnorm := max(abs(rmax), abs(rmin));
59  # ||r||_inf
60
61  printf "%4i %9.1e %9.1e %9.1e %15.7e\n",
62  K, rhok, etak, rnorm, f >> 5DNCLipopt.log;
63  close 5DNCLipopt.log;
64
65  if rnorm <= rtol then
66  { printf "Stopping: rnorm is small\n";
67    display K, rnorm; break; }
68
69  if rnorm <= etak then # update dual estimate dk;
70  save new solution
71  {let {(i,j,k,g,h) in T, (p,q,r,s,t) in T:
72      !(i=p and j=q and k=r and g=s and h=t)})
73      dk[i,j,k,g,h,p,q,r,s,t] :=
74      dk[i,j,k,g,h,p,q,r,s,t] + rhok
75      *R[i,j,k,g,h,p,q,r,s,t];
76      let {(i,j,k,g,h) in T} ck[i,j,k,g,h] :=
77      c[i,j,k,g,h];
78      let {(i,j,k,g,h) in T} yk[i,j,k,g,h] :=
79      y[i,j,k,g,h];
80      display K, etak;
81      if etak == etamin then { printf "Stopping:
82      etak = etamin\n"; break; }
83      let etak := max(etak*etafac, etamin);
84      display etak;
85  }
86  else # keep previous solution; increase rhok
87  { let {(i,j,k,g,h) in T} c[i,j,k,g,h] :=
88      ck[i,j,k,g,h];
89      let {(i,j,k,g,h) in T} y[i,j,k,g,h] :=
90      yk[i,j,k,g,h];
91      display K, rhok;
92      if rhok == rhomax then { printf "Stopping:
93      rhok = rhomax\n"; break; }
94      let rhok := min(rhok*rhofac, rhomax);

```

```

95     display rhok;
96   }
97 }
98
99 display c,y; display na, nb, nc, nd, ne, primreg,
100 rhok, etak, rnorm;
101
102 # Count how many constraint are close to being active.
103 data;
104 let nT := na*nb*nc*nd*ne; let m := nT*(nT-1);
105 let n := 2*nT;
106 let etak := 1.0001e-10;
107 printf "\n m = %8i\n n = %8i\n", m, n >>
108 5DNCLipopt.log;
109 printf "\n Constraints within tol of being
110 active\n\n" >> 5DNCLipopt.log;
111 printf "      tol      count      count/n\n" >>
112 5DNCLipopt.log;
113
114 for {K in 1..10}
115 { let kmax := card{(i,j,k,g,h) in T, (p,q,r,s,t) in T:
116                      !(i=p and j=q and k=r and g=s
117                      and h=t)
118                      and Incentive[i,j,k,g,h,p,q,r,s,t].
119                      slack <= etak};
120   printf "%9.1e %8i %8.1f\n", etak, kmax,
121   kmax/n >> 5DNCLipopt.log;
122   let etak := etak*10.0;
123 }
124 printf "Created 5DNCLipopt.log\n";

```

---

## References

1. Amestoy, Patrick R., Duff, Iain S., L'Excellent, Jean-Yves, Koster, Jacko: A fully asynchronous multifrontal solver using distributed dynamic scheduling. *SIAM J. Matrix Anal. Appl.* **23**(1), 15–41 (2001)
2. Arreckx, S., Orban D.: A regularized factorization-free method for equality-constrained optimization. Technical report GERAD G-2016-65. GERAD, Montréal, QC, Canada (2016)
3. Byrd, Richard H., Nocedal, Jorge, Waltz, Richard A.: Knitro: an integrated package for non-linear optimization. In: Di Pillo, G., Roma, M. (eds.) *Large-Scale Nonlinear Optimization*, pp. 35–59. Springer, US, Boston, MA (2006)
4. Conn, A.R., Gould, N.I.M., Toint, PhL: A globally convergent augmented Lagrangian algorithm for optimization with general constraints and simple bounds. *SIAM J. Numer. Anal.* **28**, 545–572 (1991)
5. Conn, A.R., Gould, N.I.M., Toint, PhL: LANCELOT: a fortran package for large-scale nonlinear optimization (release A). *Lecture Notes in Computational Mathematics*, vol. 17. Springer, Berlin, Heidelberg, New York, London, Paris and Tokyo (1992)
6. Fourer, R., Gay, D.M., Kernighan, B.W.: *AMPL: A Modeling Language for Mathematical Programming*, 2nd edn. Pacific Grove, Brooks/Cole (2002)

7. Friedlander, M.P., Orban, D.: A primal-dual regularized interior-point method for convex quadratic programs. *Math. Prog. Comp.* **4**(1), 71–107 (2012)
8. Friedlander, M.P., Saunders, M.A.: A globally convergent linearly constrained Lagrangian method for nonlinear optimization. *SIAM J. Optim.* **15**(3), 863–897 (2005)
9. Gill, P.E., Kungurtsev, V., Robinson, D.P.: A stabilized SQP method: global convergence. *IMA J. Numer. Anal.* **37**, 407–443 (2017)
10. Gill, P.E., Kungurtsev, V., Robinson, D.P.: A stabilized SQP method: superlinear convergence. *Math. Program. Ser. A* **163**, 369–410 (2017)
11. Gill, P.E., Murray, W., Saunders, M.A.: SNOPT: an SQP algorithm for large-scale constrained optimization. *SIAM Rev.* **47**(1), 99–131 (2005). SIGEST article
12. IPOPT open source NLP solver. <https://projects.coin-or.org/Ipopt>
13. Judd, K.L., Ma, D., Saunders, M.A., Su, C.-L.: Optimal income taxation with multidimensional taxpayer types. Working paper. Hoover Institution, Stanford University (2017)
14. KNITRO optimization software. [https://www.artelys.com/tools/knitro\\_doc/2\\_userGuide.html](https://www.artelys.com/tools/knitro_doc/2_userGuide.html)
15. LANCELOT optimization software. <http://www.numerical.rl.ac.uk/lancelot/blurb.html>
16. Murtagh, B.A., Saunders, M.A.: A projected Lagrangian algorithm and its implementation for sparse nonlinear constraints. *Math. Program. Study* **16**, 84–117 (1982)
17. NCL: <http://stanford.edu/group/SOL/multiscale/models/NCL/>
18. Robinson, S.M.: A quadratically-convergent algorithm for general nonlinear programming problems. *Math. Program.* **3**, 145–156 (1972)
19. Wächter, A., Biegler, L.T.: On the implementation of a primal-dual interior point filter line search algorithm for large-scale nonlinear programming. *Math. Program.* **106**(1) (2006)
20. Wright, S.J.: Superlinear convergence of a stabilized SQP method to a degenerate solution. *Comput. Optim. Appl.* **11**, 253–275 (1998)